

Z1013

Unterprogramm "Variablen"

Während der Programmerstellungs- und Testphase wünscht man sich oft einen Überblick über die bereits verwendeten Variablen und deren Inhalt. Das vorgestellte BASIC-Programm listet alle zur Zeit seines Aufrufs verwendeten Variablen (auch Feldvariablen) mit Ihren Namen und ihrem aktuellen Inhalt auf. Dazu ist es an das zu untersuchende Programm anzufügen und mit GOSUB 30000 zu starten. Das UP selbst benutzt nur ein Feld ZZ(8), dieses wird nicht dokumentiert und darf im Hauptprogramm nicht zur Anwendung kommen (führt zu Fehler "DD"). Für das Feld ZZ sind 43 Bytes freizuhalten. Am Ende der Abarbeitung des UPs wird dieser Speicherraum wieder freigegeben, so daß nach der Rückkehr in das Hauptprogramm das Feld ZZ(8) dort nicht "bekannt" ist. Das Programm ist für den 10-KByte-BASIC-Interpreter des Z1013 geschrieben. Zur Anpassung an andere KC-Typen ist hier die Bedeutung der POKE-Adressen ausgegeben:

```
11105 = 2B61H, Beginn Eingabepuffer
11227 = 2BDBH, erste freie Adresse hinter den Variablenlisten
11223 = 2BD7H, Adresse der Liste von Einfach- und Stringvariablen
11225 = 2BD9H, Adresse der Feldvariablenliste
11107 = 2B63H, Beginn Eingabepuffer + 2.
```

Das Feld ZZ wird folgendermaßen eingesetzt:

```
ZZ(1) = Beginn Variablen
ZZ(2) = Beginn Felder
ZZ(3) = Ende Felder
ZZ(4) = Variablenzähler für PAUSE
ZZ(5) = Real oder String/LEN (String)
ZZ(6) = Zähler für vier Bytes Real/ADR (String)
ZZ(7) = Längenzähler für String
ZZ(8) = Hilfszähler für Indizierung
ZZ(0) = Ausgabe Real
```

```
30000 DOKE 11105, DEEK(11227)
30001 DIM ZZ(8)
30002 ZZ(1)=DEEK(11223):ZZ(2)=DEEK(11225):ZZ(3)=DEEK(11105):ZZ(4)=1
30003 IF ZZ(4)/10=INT(ZZ(4)/10) THEN PRINT "Weiter mit <S4/F>";:
    PAUSE:PRINT
30004 PRINT CHR$(PEEK(ZZ(1)+1));:ZZ(5)=PEEK(ZZ(1))
30005 IF ZZ(5)<128 THEN PRINT CHR$(ZZ(5));:GOSUB 30100:GOTO 30007
30006 PRINT CHR$(ZZ(5)-128);"$";:GOSUB 30200
30007 ZZ(1)=ZZ(1)+6:ZZ(4)=ZZ(4)+1:IF ZZ(1)<ZZ(2) THEN 30003
30008 ZZ(1)=PEEK(ZZ(2)+4)*2+ZZ(2)+3
30009 ZZ(8)=0
30010 DOKE ZZ(8)*2+11107,0:ZZ(8)=ZZ(8)+1
30011 IF ZZ(8)<PEEK(ZZ(2)+4) THEN 30010
30012 IF ZZ(4)/10=INT(ZZ(4)/10) THEN PRINT "Weiter mit <S4/F>";:
    PAUSE:PRINT
30013 PRINT CHR$(PEEK(ZZ(2)+1));
30014 IF PEEK(ZZ(2))<128 THEN PRINT CHR$(PEEK(ZZ(2)));:GOTO 30016
30015 PRINT CHR$(PEEK(ZZ(2))-128);"$";
30016 PRINT "(";:ZZ(8)=PEEK(ZZ(2)+4)-1
30017 PRINT RIGHT$(STR$(DEEK(ZZ(8)*2+11107)),LEN(STR$(DEEK(ZZ(8)*2
    +11107)))-1);
30018 PRINT ",";:ZZ(8)-1:IF ZZ(8)>=0 THEN 30017
30019 PRINT CHR$(8);")";:IF PEEK(ZZ(2))<128 THEN GOSUB 30100:ELSE
    GOSUB 30200
```

```

30020 ZZ(1)=ZZ(1)+4:ZZ(4)=ZZ(4)+1:ZZ(8)=PEEK(ZZ(2)+4)-1
30021 IF DEEK(ZZ(8)*2+11107)<DEEK(ZZ(8)*2+ZZ(2)+5)-1 THEN 30023
30022 DOKE ZZ(8)*2+11107,0:GOTO 30024
30023 DOKE ZZ(8)*2+11107, DEEK(ZZ(8)*2+11107)+1:GOTO 30012
30024 ZZ(8)=ZZ(8)-1:IF ZZ(8)>0 THEN 30021
30025 ZZ(2)=ZZ(2)+DEEK(ZZ(2)+2)+4
30026 IF ZZ(2)<ZZ(3) THEN 30008
30027 DOKE 11227, DEEK(11105):RETURN
30100 ZZ(6)=0
30101 POKE ZZ(3)+7+ZZ(6), PEEK(ZZ(1)+2+ZZ(6))
30102 ZZ(6)=ZZ(6)+1:IF ZZ(6)<4 THEN 30101
30103 PRINT TAB(5);"="; ZZ(0):RETURN
30200 ZZ(5)=PEEK(ZZ(1)+2):PRINT TAB(5);"=";:ZZ(6)=DEEK(ZZ(1)+4):
PRINT CHR$(34);
30201 ZZ(7)=-1:GOTO 30203
30202 PRINT CHR$(PEEK(ZZ(6)+ZZ(7)));
30203 ZZ(7)=ZZ(7)+1:IF ZZ(7)<ZZ(5) THEN 30202
30204 PRINT CHR$(34);"(";RIGHT$(STR$(ZZ(5)),LEN(STR$(ZZ(5)))-1;")":
RETURN

```

P. Rusche

Z1013

Nutzung der I/O-Adresse 4

Die Ein-/Ausgabeadresse 4 ist beim Z1013 für Hardwareschaltfunktionen vorgesehen. Bei Rechnerkonfigurationen mit Baugruppenträger treten beim Lesen von dieser Adresse Probleme auf - der Rechner liest ständig FFH. Die Ursache liegt darin, daß vom Baugruppenträger bestimmte I/O-Bereiche der Grundausbaustufe gesperrt werden (Gatter A1.1 im Stromlaufplan [1]). Für die Adressen 4 bis 7 (/IOSEL1) läßt sich diese Erscheinung beseitigen, indem der Leiterzug von A4.1, Pin 2 nach A1.1, Pin 1 auf der Leiterplatte des Baugruppenträgers durchgekratzt wird. Bei diesem Eingriff erlischt allerdings der Garantieanspruch, da die Baugruppe beim Hersteller und im Service nicht mehr automatisch prüfbar ist!

T. Sander

Literatur

[1] MRB Z1013 auf einen Blick, FUNKAMATEUR 38 (1989), H.6, S.289